```
Clear["Global`*"]
```

MD Benchmark Code

The program is divided in four chapters:

1. REBO Potentials.
In this chapter the form of the REBO potential to be tested is assigned. In the section "2nd generation Brenner potential", the default setting for this potential is implemented, according to [2]; in particular, in the subsection "Potential components" the components introduced in (17) are specified. In the section "Analytical discrete model" the definition of the nanostresses (19) is implemented; this definition is independent of the REBO potential one choses.

2. Armchair CNTs
In this chapter the equilibrium problem for armchair CNTs is solved. In the section "Generalities" the geometric conditions on the order parameters are established and the nanostresses are computed. In the section "Solution of the equilibrium equations" the solution of the system 3_1 and 4_1 is determined as a function of the applied force F and the chiral number n. In the section "Radius" the natural radius is computed as a function of F and n and then determined for F=0, namely in the natural configuration. In the section "Energy" the natural energy is computed as a function of F and n and then determined for F=0, namely the cohesive energy. In the section "Young's modulus" the current and the referential lengths of a CNT are determined, and the strain measure is defined, as a function of F and n; on introducing the nominal thickness, the Young's modulus is defined as a function of F and n, and then computed for a tiny value of F, up to convergence. In section "Poisson coefficient", the named material parameter is defined as a function of F and n, and then computed for a tiny value of F, up to convergence.

3. Zigzag CNTs.
This chapter has the same section as the previous one, but implemented for the zigzag case; the different geometric constraints are properly included.

4. Summary of Results.
In this chapter the benchmark solutions are collected for the visualization in the CDF file Benchmark_solutions.cdf.

Equation numbers and bibliographic reference numbers are those of the article associated to this code.

1. REBO Potentials

2nd generation Brenner potential

G Function

According to the definition in reference [2]

```
d1={Subscript[d, 0]-> 0.375449999999999,Subscript[d, 1]->1.406776475152104,
Subscript[d, 2]->2.254377494462425,Subscript[d, 3]->2.031282890266629,
Subscript[d, 4]->1.429711740681566,Subscript[d, 5]->0.502401399437276};
G1[θ_]=∑_{i=0}^5 d_i*Cos[θ]^i/.d1;

d2={Subscript[d, 0]->0.707277245734129,Subscript[d, 1]->5.677435848489921,
Subscript[d, 2]->24.09701877750397,Subscript[d, 3]->57.59183195999613,
Subscript[d, 4]->71.88287000288395,Subscript[d, 5]->36.27886067346856};
G2[θ_]=∑_{i=0}^5 d_i*Cos[θ]^i/.d2;

d3={Subscript[d, 0]->-0.644399999999948,Subscript[d, 1]->-6.207999999999618,
Subscript[d, 2]->-20.05899999999891,Subscript[d, 3]->-30.22799999999852,
Subscript[d, 4]->-21.72399999999905,Subscript[d, 5]->-5.990399999999763};
G3[θ_]=∑_{i=0}^5 d_i*Cos[θ]^i/.d3;
G[θ_]=Piecewise[{{G1[θ],0<=θ<0.6082*π},{G2[θ],0.6082*π<=θ<2/3*π},{G3[θ],2/3*π<=θ<=π}}];
```

Potential parameters

```
ev=0.1602176487;(*unit conversion constant*)
```

```
potparameters={Subscript[B, 1]-> 12388.79197798,Subscript[B, 2]->17.56740646509,
Subscript[B, 3]->30.71493208065, Subscript[β, 1]->4.7204523127,Subscript[β, 2]->1.4332132499,
Subscript[β, 3]->1.3826912506,Q->0.3134602960833,A->10953.544162170,α0->4.7465390606595};
```

```
T=-0.004048375;
```

Potential components

equation (17)
(reference [2])

```
VA[r_]=-∑_{n=1}^3 B_n*E^(-β_n*r)/.potparameters;
VR[r_]=(1+Q/r)*A*E^(-α0*r)/.potparameters;
DVA[r_]=D[VA[r],r];
DVR[r_]=D[VR[r],r];
ba=(1+2*G[β])^(-(1/2))+2*T*(1-Cos[Θ1]^2);
bb=(1+G[α]+G[β])^(-(1/2))+T*(2*(1-Cos[Θ2]^2)+(1-Cos[Θ3]^2)+(1-Cos[Θ4]^2));
```

Analytical discrete model

Geometry

```
n1[n_]:=2*n;
n2[n_]:=n;
```

Nanostresses

Equation (19)

```
σa=DVR[a]+ba*DVA[a];
σb=DVR[b]+bb*DVA[b];
τα=D[bb,α]*VA[b];
τβ=1/4*(D[ba,β]*VA[a]+2*D[bb,β]*VA[b]);
T1=1/2*D[ba,Θ1]*VA[a];
T2=1/2*D[bb,Θ2]*VA[b];
T3=D[bb,Θ3]*VA[b];
T4=D[bb,Θ4]*VA[b];
```

2. Armchair CNTs

Generalities

Geometric constraints

Reference [1]

```
ϕ[n_]:=π/n1[n];
```

```
β[n_]:=ArcCos[-Cos[α/2]*Cos[ϕ[n]]];
Θ1[n_]:=2*ArcSin[(Cos[α/2]*Sin[ϕ[n]])/Sin[β[n]]];
Θ2[n_]:=ArcSin[Sin[ϕ[n]]/Sin[β[n]]];
Θ3[n_]:=2*Θ2[n];
Θ4[n_]:=0;
```

```
σan[n_]:=σa/.{β->β[n],Θ1->Θ1[n],Θ2->Θ2[n],Θ3->Θ3[n],Θ4->Θ4[n]}
σbn[n_]:=σb/.{β->β[n],Θ1->Θ1[n],Θ2->Θ2[n],Θ3->Θ3[n],Θ4->Θ4[n]}
ταn[n_]:=τα/.{β->β[n],Θ1->Θ1[n],Θ2->Θ2[n],Θ3->Θ3[n],Θ4->Θ4[n]}
τβn[n_]:=τβ/.{β->β[n],Θ1->Θ1[n],Θ2->Θ2[n],Θ3->Θ3[n],Θ4->Θ4[n]}
T1n[n_]:=T1/.{β->β[n],Θ1->Θ1[n],Θ2->Θ2[n],Θ3->Θ3[n],Θ4->Θ4[n]}
T2n[n_]:=T2/.{β->β[n],Θ1->Θ1[n],Θ2->Θ2[n],Θ3->Θ3[n],Θ4->Θ4[n]}
T3n[n_]:=T3/.{β->β[n],Θ1->Θ1[n],Θ2->Θ2[n],Θ3->Θ3[n],Θ4->Θ4[n]}
T4n[n_]:=T4/.{β->β[n],Θ1->Θ1[n],Θ2->Θ2[n],Θ3->Θ3[n],Θ4->Θ4[n]}
```

Solution of the equilibrium equations

```
solA[F_,n_]:=FindRoot[{σan[n]==0,σbn[n]-F/n1[n]*Sin[α/2]==0,ταn[n]+
2*D[β[n],α]*τβn[n]+D[Θ1[n],α]*T1n[n]+2*D[Θ2[n],α]*T2n[n]+D[Θ3[n],α]*T3n[n]
-1/2*F/n1[n]*b*Cos[α/2]==0},{{a,1.4},{b,1.4},{α,2}}]
```

Radius

Reference [1]

```
ρa[n_]:=(b/2*Cos[α/2]+a/2*Cos[ϕ[n]])/Sin[ϕ[n]];
ρ[n_]:=Sqrt[ρa[n]^2+a^2/4];
rA[F_,n_]:=ρ[n]/10/.solA[F,n];(*radius in nm*)
```

Natural radius (output in Table 3)

```
ρA[n_]:=rA[0,n]
```

Energy

```
Va=VR[a]+ba*VA[a];
Vb=VR[b]+bb*VA[b];
```

```
V[F_,n_]:=(Va+2*Vb)/2/.{β->β[n],Θ1->Θ1[n],Θ2->Θ2[n],Θ3->Θ3[n],Θ4->Θ4[n]}/.solA[F,n]
```

Cohesive energy (output of Table 3)

Reference [1]

```
EnergyA[n_]:=V[0,n](*ev/atom*)
```

Young Modulus

Equations (13)-(14)

```
λA[n_]:=2*Sin[α/2]*n2[n]*b;(*current length*)
λ0A[n_]:=λA[n]/.solA[0,n];(*referential length*)
ϵA[n_]:=(λA[n]-λ0A[n])/λ0A[n];(*strain measure*)
eA[F_,n_]:=ϵA[n]/.solA[F,n];
```

```
t=0.34(*nm*);(*nominal thickness*)
```

```
YA[F_,n_]:=(F/eA[F,n])*ev*(1/(2*π*rA[F,n]*t))*10(*Young modulus, GPa*)
```

Young modulus in the origin (output in Table 4)

```
YoungA[n_]:=If[n<20,YA[10^-10,n],YA[10^-9,n]](*GPa*)
```

Poisson coefficient

Equation (15)

```
νA[F_,n_]:=-((rA[F,n]-rA[0,n])/rA[0,n])*1/eA[F,n]
```

Poisson coefficient in the origin (output in Table 4)

```
PoissonA[n_]:=If[n<20,νA[10^-10,n],νA[10^-9,n]]
```

3. Zigzag CNTs

Generalities

Geometric constraints

Reference [1]

```
ϕz[n_]:=π/n;
```

```
βz[n_]:=π-ArcSin[Sin[α/2]/Cos[ϕz[n]/2]];
Θ1z[n_]:=ϕz[n];
Θ2z[n_]:=ArcSin[(Sin[βz[n]]*Sin[ϕz[n]])/Sin[α]];
Θ3z[n_]:=0;
Θ4z[n_]:=2*Θ2z[n];
```

```
σaz[n_]:=σa/.{β->βz[n],Θ1->Θ1z[n],Θ2->Θ2z[n],Θ3->Θ3z[n], Θ4->Θ4z[n]}
σbz[n_]:=σb/.{β->βz[n],Θ1->Θ1z[n],Θ2->Θ2z[n],Θ3->Θ3z[n], Θ4->Θ4z[n]}
ταz[n_]:=τα/.{β->βz[n],Θ1->Θ1z[n],Θ2->Θ2z[n],Θ3->Θ3z[n], Θ4->Θ4z[n]}
τβz[n_]:=τβ/.{β->βz[n],Θ1->Θ1z[n],Θ2->Θ2z[n],Θ3->Θ3z[n], Θ4->Θ4z[n]}
T1z[n_]:=T1/.{β->βz[n],Θ1->Θ1z[n],Θ2->Θ2z[n],Θ3->Θ3z[n], Θ4->Θ4z[n]}
T2z[n_]:=T2/.{β->βz[n],Θ1->Θ1z[n],Θ2->Θ2z[n],Θ3->Θ3z[n], Θ4->Θ4z[n]}
T3z[n_]:=T3/.{β->βz[n],Θ1->Θ1z[n],Θ2->Θ2z[n],Θ3->Θ3z[n], Θ4->Θ4z[n]}
T4z[n_]:=T4/.{β->βz[n],Θ1->Θ1z[n],Θ2->Θ2z[n],Θ3->Θ3z[n], Θ4->Θ4z[n]}
```

Solution of the equilibrium equations

```
solZ[F_,n_]:=FindRoot[{σaz[n]-F/n2[n]==0,σbz[n]+F/(2*n2[n])*Cos[βz[n]]==0,
ταz[n]+2*D[βz[n],α]*τβz[n]+2*D[Θ2z[n],α]*T2z[n]+D[Θ4z[n],α]*T4z[n]
-1/2*F/n2[n]*b*D[βz[n],α]*Sin[βz[n]]==0},{{a,1.4},{b,1.4},{α,2}}]
```

Radius

Reference [1]

```
rz[n_]:=Sin[βz[n]]/(2*Sin[ϕz[n]/2])*b;
r[F_,n_]:=rz[n]/10/.solZ[F,n];(*radius in nm*)
```

Natural radius (output in Table 3)

```
ρZ[n_]:=r[0,n](*nm*)
```

Energy

```
VZ[F_,n_]:=(Va+2*Vb)/2/.{β->βz[n],Θ1->Θ1z[n],Θ2->Θ2z[n],Θ3->Θ3z[n], Θ4->Θ4z[n]}/.solZ[F,n]
```

Cohesive energy (output in Table 3)

```
EnergyZ[n_]:=VZ[0,n](*ev/atom*)
```

Young Modulus

Equations (13)-(14)

```
λZ[n_]:=(1-b/a*Cos[βz[n]])*n1[n]*a;(*current length*)
λ0Z[n_]:=λZ[n]/.solZ[0,n];(*referential length*)
∊Z[n_]:=(λZ[n]-λ0Z[n])/λ0Z[n];(*strain measure*)
eZ[F_,n_]:=∊Z[n]/.solZ[F,n];
```

```
YZ[F_,n_]:=(F/eZ[F,n])*ev*(1/(2*π*r[F,n]*t))*10(*Young modulus,GPa*)
```

Young modulus in the origin (output in Table 4)

```
YoungZ[n_]:=YZ[10^-8,n](*GPa*)
```

Poisson ratio

Equation (15)

```
ν[F_,n_]:=-((r[F,n]-r[0,n])/r[0,n])*1/eZ[F,n]
```

Poisson coefficient in the origin (output in Table 4)

```
PoissonZ[n_]:=ν[10^-8,n]
```

4. Summary of Results

```
Radius[0,n_]:=ρA[n]
Radius[1,n_]:=ρZ[n]
Energy[0,n_]:=EnergyA[n]
Energy[1,n_]:=EnergyZ[n]
Young[0,n_]:=YoungA[n]
Young[1,n_]:=YoungZ[n]
Poisson[0,n_]:=PoissonA[n]
Poisson[1,n_]:=PoissonZ[n]
```

Creation of the shell for cdf file

```
Manipulate[
Column[{ NumberForm[Radius[q,n],4]  "radius [nm]:",NumberForm[Energy[q,n],5] "cohesive energy
 NumberForm[Young[q,n],7] "Young modulus [GPa]:", NumberForm[Poisson[q,n],4] "Poisson coeffici
Left,", "],
{{q,0,"show"},{0->"armchair",1->"zigzag"},Setter},
{n,3,30},
AutorunSequencing->{1,2},
TrackedSymbols:>{q,n},
ControlPlacement->Top,
SaveDefinitions->True,
ControlType->InputField]
```